Создание анимационных изображений на языке объектно-ориентированного программирования
Visual Basic

 Введение. Анимация
Справочная литература:

· Информатика и информационные технологии Учебник для 10-11 классов./Угринович Н.Д, §4.10, Анимация, проект «Движение круга».

· Практикум по информатике и информационным технологиям. Учебное пособие для общеобразовательных учреждений/Н.Д. Угринович, Л.Л. Босова, Н.И. Михайлова. §5.11, проекты «Движение Земли вокруг Солнца», «Мультимедиа».

В детстве каждому доводилось смотреть мультипликационные фильмы. Сейчас их также называют анимацией. В последнее время анимация нашла широкое применение в Интернете, в первую очередь, в рекламных банерах. Анимация привлекает внимание гораздо сильнее, чем статические изображения, на небольшой площади можно поместить много информации за счет чередующихся кадров.

Под анимацией понимают отображение в быстром темпе последовательности кадров, каждый из которых несколько отличается от соседних. Кадр демонстрируется в течение определенного промежутка времени. При достаточно высокой частоте смены кадров создается иллюзия движения.

До недавнего времени анимационные фильмы рисовались на бумаге. Для создания десятиминутного мультфильма требовалось нарисовать 10 тысяч изображений. В последние годы анимационные технологии переведены на компьютерную основу, что позволило резко ускорить создание таких фильмов. И хотя концепции разработки анимации остались прежними, используемые методы значительно отличаются. Качество анимации определяется числом используемых кадров. Чем больше кадров содержит анимация на единицу времени, тем более плавным и сглаженным будет движение в процессе проигрывания.
В современном «индустриальном» программировании обычно уже не ставится вопрос о том, как осуществить движение графических объектов или запрограммировать поворот трехмерной фигуры. Есть готовые стандартные библиотеки (DirectX и OpenGL, к примеру), есть библиотеки для удобной работы с этими библиотеками, есть даже готовые специализированные среды разработки (в частности, для разработки игр предназначены DIV Games и Studio Dark Basic, очень популярна программа для создания сложной интерактивной анимации Macromedia Flash MX) и т.д. Скорее всего, придется иметь дело хотя бы с некоторыми из этих вещей; тем не менее, интересно и полезно «заглянуть за кулисы», узнать, как это происходит «на самом деле», а не просто вызвать готовую функцию. К тому же (и это более важно) существуют концепции и приемы, которые не зависят ни от конкретной библиотеки, ни даже от операционной системы.
Для создания анимационного фильма в программе Visual Basic достаточно подготовить несколько изображений или кадров, расположить в нужном порядке и указать скорость смены кадров. Другой способ создания ролика заключается в автоматическом перемещении графического объекта. Для подготовки такого фильма следует создать изображение, указать число кадров в анимационном ролике, величину смещения объекта между кадрами, угол поворота при вращении рисунка, изменение размеров объекта и другие параметры.

Справочная литература:

· Информатика и информационные технологии Учебник для 10-11 классов./Угринович Н.Д, §4.10, Анимация, проект «Движение круга».

· Практикум по информатике и информационным технологиям. Учебное пособие для общеобразовательных учреждений/Н.Д. Угринович, Л.Л. Босова, Н.И. Михайлова. §5.11, проекты «Движение Земли вокруг Солнца», «Мультимедиа».

Покадровая анимация

Покадровая анимация заключается в смене кадров, т.е. в объект Image /или Picture / в строгой последовательности загружаются файлы рисунков, которые создаются в графическом редакторе. Каждый кадр должен незначительно отличаться от предыдущего. Это сложное и кропотливое дело. Необходимо, чтобы пользователь обладал  художественными способностями и  в достаточной степени владел навыками работы в среде графического редактора. Поэтому для создания кадров можно использовать Gif-анимацию.

В Интернете можно найти движущиеся  рисунки с расширением gif, которые выглядят как  анимация. Эти графические файлы представляют собой несколько рисунков, которые выводятся на экран по очереди. Специальные программы gif-аниматоры помогут разделить файл на отдельные кадры и записать каждый кадр в отдельный файл. Эти файлы можно использовать для создания анимации в Visual Basic. Разбиение файла на кадры можно осуществить с помощью программы Adobe ImageReady.
Массив объектов

Массивом объектов называется группа объектов, имеющих одинаковое имя (свойство Name), тип и общую процедуру обработки событий. Элементы в массиве пронумерованы по порядку. Порядковый номер элемента называется индексом и записывается в скобках сразу после имени массива. Объекты, входящие в массив, имеют свойство Index. Это свойство равно порядковому номеру элемента массива. Использование массивов объектов позволяет значительно сократить программный код.
В проекте Птичка для размещения кадров анимированного изображения использовался массив класса Image.

Для создания массива необходимо:

· Создать объект Image;

· Скопировать объект в буфер обмена, нажав CTRL+C
· Вставить объект из буфера CTRL+V столько раз, сколько кадров
Алгоритм и программный код мини-анимации.

В каждый элемент массива вставить соответствующий анимационный кадр, изменив свойство Picture. Затем разместить на форме объект Timer  и написать программный код для создания анимации. В объект Image1 по очереди загружать кадры из массива объектов Image2
Программный код мини-анимации. 
Private Sub Timer1_Timer()

Picture1.Picture = Image1(a).Picture

a = a + 1

If a = 7 Then a = 0

End Sub
Анимация, основанная на изменении свойств объекта 

Положение любого объекта на форме определяется значением двух его свойств – Top и Left. Свойство Top равно расстоянию от верхнего левого угла объекта до верхней границы формы в тех единицах измерения, которые заданы свойством ScaleMode. Свойство Left равно расстоянию от верхнего левого угла объекта до левой границы формы.

Если во время работы программы менять значения свойства Top объекта, он будет двигаться вверх или вниз.

 Если менять значения свойства Left – вправо или влево. 
Если менять значения обоих свойств, тело будет перемещаться по более сложной траектории.
Эффект увеличения или уменьшения объекта

 Для создания данной анимации объекта необходимо изменять его размеры, то есть свойства Width /ширина/ и Height /высота/. 
Примером такой анимации служит проект Футбол. 
Image1.Width = Image1.Width + 2

Image1.Height = Image1.Height + 2
· Смотрите электронный сборник проектов
Моделирование  движения объектов и по заданной траектории. 
 Справочная литература:

· Информатика и информационные технологии Учебник для 10-11 классов./Угринович Н.Д, §4.10, проект «Построение графика функции».

· Практикум по информатике и информационным технологиям. Учебное пособие для общеобразовательных учреждений/Н.Д. Угринович, Л.Л. Босова, Н.И. Михайлова. §5.10, проект 5.13.

Например, движение по параболе в проекте Модели обеспечивается следующим образом:
For i = -2 To 2 Step 0.00001

Shape1.Left = i
Shape1.Top = i ^ 2

Next i

Движение по синусоиде:

For i = -6.28 To 6.28 Step 0.00005

Shape1.Left = i

Shape1.Top = Sin(i)

Next i

· Смотрите электронный сборник проектов. Проект Модели.
Движение по окружности.
Движение по окружности задают формулы:
X = R * Cos (,  Y = R * Sin (, где R - радиус окружности; ( - угол в радианах, который изменяется во время работы программы.

Движение по случайной траектории

Для задания случайной траектории можно использовать датчик случайных чисел – функцию RND. 
 x = Rnd * 10

y = Rnd * 10

Shape1.Left = x

Shape1.Top = y
Поточечная анимация
Поточечная анимация основана на окрашивании одного пикселя экрана в заданный цвет. В проекте Анимация по точкам  это записывается следующим программным кодом:
For x = 1 To Picture3.Width Step Screen.TwipsPerPixelX

      For y = 1 To Picture3.Height Step Screen.TwipsPerPixelY

              color = Picture3.Point(x, y)

              If color <> RGB(255, 255, 255) Then

           Picture2.PSet (x, y), color

         End If

      Next y

   Next x

Пользовательское меню. Menu Editor
Для вызова редактора меню:

· Инструменты( Редактор меню

· [image: image12.png]


Ctrl+E

· Кнопка на стандартной панели

На появившейся диалоговой панели Menu Editor создать заголовок первого уровня. В поле Название внести надпись, например Файл, в поле Имя - имя объекта, например Mnf.

Для создания следующего пункта меню щелкнуть по кнопке Дальше.

Для создания пункта меню уровня команд, например, команды меню Файл, щелкнуть по кнопке со стрелкой вправо. Ввести надпись Очистка и имя объекта MnCls. Для перехода на верхний уровень меню используется кнопка со стрелкой влево, на нижний уровень – кнопка со стрелкой вправо, для смены последовательности команд используются кнопки со стрелками вниз и вверх, для вставки пункта меню кнопка Вставить, для удаления – Удалить, выхода из редактора меню ОК.
В режиме конструктора проекта раскрыть меню и для каждого пункта меню создать событийную процедуру, которая будет выполняться по щелчку мыши.

Например, для очистки объекта 

Private Sub MnCls_Click()

Picture1.Cls

End Sub
Private Sub MnEnd_Click()

End

End Sub
Как подключить звуковой файл с расширением wav.

Подготовительный этап. Продумать сценарий программы.  Для этого необходимо ответить для себя на вопросы: Какие необходимо объекты? При каком событии будет происходить воспроизведение звука в проекте? Какую музыку будете использовать? Необходимо ли самостоятельно подготовить звуковой файл/файл «озвучки»/ с помощью микрофона?
Подключить функцию воспроизведения звукового файла /стандартная библиотека Windows winmm.dll/
Private Declare Function PlaySound Lib "winmm.dll" _

Alias "PlaySoundA" (ByVal lpszSoundName As String, _

ByVal hModule As Long, ByVal dwFlags As Long) As Long

Const SND_ASYNC = &H1

Const SND_SYNC = &H0

Const SND_LOOP = &H8

В папку проекта скопировать звуковой файл или создать свой с расширением wav.
В событийную процедуру объекта написать программный код вызова функции воспроизведения звука, указав имя файла

      Call PlaySound(CStr(CurDir) + "\имя.wav", 0, _

                        SND_FILENAME Or SND_ASYNC Or SND_LOOP)

В событийную процедуру выхода из проекта остановить воспроизведение звука, написав следующий программный код
Call PlaySound(0, 0,  SND_ASYNC)

Пример. 

 ' Объявление функции воспроизведения звукового файла

Private Declare Function PlaySound Lib "winmm.dll" _

Alias "PlaySoundA" (ByVal lpszSoundName As String, _

ByVal hModule As Long, ByVal dwFlags As Long) As Long

Const SND_ASYNC = &H1

Const SND_SYNC = &H0

Const SND_LOOP = &H8
При нажатии на кнопку звук Command1 происходит воспроизведение звукового файла 1. wav
Private Sub Command1_Click()

      Call PlaySound(CStr(CurDir) + "\1.wav", 0, _

                        SND_FILENAME Or SND_ASYNC Or SND_LOOP)

End Sub

При событии Click объекта Command3  /кнопка ку-ку на форме/ воспроизводится звук из файла online.wav

 Private Sub Command3_Click()

Call PlaySound(CStr(CurDir) + "\online.wav", 0, _

                        SND_FILENAME Or SND_ASYNC Or SND_LOOP)

End Sub

При щелчке по кнопке выход Command2 осуществляется выключение звука и выход из программы

Private Sub Command2_Click()

Call PlaySound(0, 0, _

                         SND_ASYNC)

  End

End Sub

Общие сведения о программе звукозаписи

Программа «Звукозапись» предназначена для записи, смешивания, воспроизведения и редактирования звукозаписей. Кроме этого, программа звукозаписи позволяет связывать звуки с другим документом или вставлять их в него. Изменение несжатой звукозаписи может выполняться следующими способами: 

· добавление звуков в файл; 

· удаление части звукозаписи; 

· изменение скорости воспроизведения; 

· изменение громкости воспроизведения; 

· изменение направления воспроизведения; 

· изменение или преобразование типов звукозаписей; 

· добавление эха. 

Примечания

Для записи звуков к компьютеру следует подключить микрофон. 

Записанные звуки сохраняются в формате звукозаписи (WAV). 

Для получения сведений об использовании программы «Звукозапись» обращайтесь к меню Справка этой программы.
 Проект ПТИЧКА
Private Declare Function PlaySound Lib "winmm.dll" _

Alias "PlaySoundA" (ByVal lpszSoundName As String, _

ByVal hModule As Long, ByVal dwFlags As Long) As Long' функция воспроизведения звукового файла
Const SND_ASYNC = &H1

Const SND_SYNC = &H0

Const SND_LOOP = &H8

Dim a As Integer

' воспроизведение звука
Private Sub Command1_Click() ‘при щелчке на кнопку1 загружается звуковой файл 1.wav
      Call PlaySound(CStr(CurDir) + "\1.wav", 0, SND_FILENAME Or SND_ASYNC Or SND_LOOP)

End Sub

Private Sub Command2_Click()‘ выход из программы
Call PlaySound(0, 0, SND_ASYNC)

  End

End Sub

Private Sub Command3_Click()при щелчке на кнопку3 загружается звуковой файл online.wav
Call PlaySound(CStr(CurDir) + "\online.wav", 0, SND_FILENAME Or SND_ASYNC Or SND_LOOP)

End Sub

Private Sub Form_Load()

a = 0

 End Sub

Private Sub Timer1_Timer() ‘процедура смены кадров
Image1.Picture = Image2(a).Picture

a = a + 1

If a = 7 Then a = 0  ‘если загружен последний восьмой кадр, то а=0 начать процедуру загрузки заново с первого кадра
End Sub
Проект СЛОНЕНОК
Private Sub Form_Load()

k = 0

End Sub

Private Sub Command1_Click()
End

End Sub

Private Sub Timer1_Timer()

Image1.Picture = Image2(k).Picture

k = k + 1

If k = 8 Then k = 0

End Sub
Проект ДОРОГА
Private Sub Timer1_Timer()

Picture3.Top = Picture3.Top + 15

If Picture3.Top > Picture2.Height Then

Picture3.Top = -Picture2.Height

End If

End Sub

Проект ГОНКИ

Dim h1 As Integer

Dim h2 As Integer


Private Sub Command1_Click()

Timer1.Enabled = True

Timer2.Enabled = True

End Sub


Private Sub Command2_Click()

Picture3.Left = 5400

Picture2.Left = 5400

Label1.Caption = "ГОНКИ"

End Sub


Private Sub Command3_Click()

End

End Sub

Private Sub Timer1_Timer()

Picture2.Left = Picture2.Left - h1

If Picture2.Left < Picture1.Left Then

Timer2.Enabled = False

Timer1.Enabled = False

Label1.Caption = "Синий пришел первым»
Label1.ForeColor = vbBlue

Picture2.Left = Picture1.Left

End If

End Sub

Private Sub Timer2_Timer()

Picture3.Left = Picture3.Left - h2

If Picture3.Left < Picture1.Left Then

Timer1.Enabled = False

Timer2.Enabled = False

Label1.Caption = "Красный пришел первым"

Label1.ForeColor = vbRed

Picture3.Left = Picture1.Left

End If

End Sub

Private Sub Timer3_Timer()

h1 = Rnd * 25 + 1

h2 = Rnd * 25 + 1

End Sub
Проект ПОДСОЛНУХ
Dim fi As Single

Dim s As Integer


Private Sub Form_Load()

fi = 3.14

End Sub


Private Sub Timer1_Timer()

fi = fi - 0.1

If fi < 0 Then

Picture1.Left = Picture1.Left + Rnd * 150

Picture1.Top = Picture1.Top - Rnd * 100

Else

Picture1.Left = 2500 * Cos(fi) + 3350

Picture1.Top = -2500 * Sin(fi) + 3000

End If

End Sub

Private Sub Timer2_Timer()

 Image2.Top = Image2.Top + 50

 Image2.Left = Image2.Left + 35

If Image2.Top > 2500 Then Timer2.Enabled = False

End Sub
Проект МОДЕЛЬ 
Private Sub MnAbs_Click()’ движение по графику модуля y = | x |

Shape1.Visible = True

Picture1.Scale (-2, 2.5)-(2, -0.5)

For i = -2 To 2 Step 0.00001

Shape1.Left = i

Shape1.Top = Abs(i)

Next i

For x = -2 To 2 Step 0.01

y = Abs(x)

Picture1.PSet (x, y), vbGreen

Next x

End Sub

Private Sub MnCls_Click()’очистка Picture1

Picture1.Cls

Shape1.Visible = False

End Sub

Private Sub MnEnd_Click()’ выход из программы
End

End Sub

Private Sub MnOcr_Click()’ движение по окружности
Shape1.Visible = True

Picture1.Scale (-1.5, 1.5)-(1.5, -1.5)

For i = 0 To 6.28 * 5 Step 0.0001

x = Cos(i)

y = Sin(i)

Shape1.Left = x

Shape1.Top = y

Next i

For i = 0 To 6.28 * 5 Step 0.0001

x = Cos(i)

y = Sin(i)

Picture1.PSet (x, y), vbGreen

Next i

End Sub

Private Sub MnPar_Click()’ движение по параболе y = x2

Shape1.Visible = True

Picture1.Scale (-2, 4)-(2, -0.5)

For i = -2 To 2 Step 0.00001

Shape1.Left = i

Shape1.Top = i ^ 2

Next i

For x = -2 To 2 Step 0.01

y = x ^ 2

Picture1.PSet (x, y), vbGreen

Next x

End Sub

Private Sub MnRnd_Click()’случайное движение 
Shape1.Visible = True

Picture1.Scale (0, 10)-(10, 0)

For i = 0 To 15

x = Rnd * 10

y = Rnd * 10

Shape1.Left = x

Shape1.Top = y

For k = 1 To 500000

Next k

Picture1.Line -(x, y), vbGreen

Next i

End Sub

Private Sub MnSin_Click()’ движение по синусоиде y = sin x
Picture1.Scale (-6.28, 2.5)-(6.28, -2.5)

Shape1.Visible = True

Shape1.Left = Form1.Left

Shape1.Top = Form1.Top + Form1.Height / 2

For i = -6.28 To 6.28 Step 0.00005

Shape1.Left = i

Shape1.Top = Sin(i)

Next i

For x = -6.28 To 6.28 Step 0.01

y = Sin(x)

Picture1.PSet (x, y), vbGreen

Next x

End Sub
Дополнительные задачи
Проект ИГРА
Dim dmin As Integer

Dim df As Integer

Private Sub Check()

If ShpCircle.Left = ShpSquare.Left And ShpCircle.Top = ShpSquare.Top Then

Label1.Caption = "Ура,вы попали в цель! Количество очков " & (100 - (df - dmin))

TimerUp.Enabled = False
TimerDown.Enabled = False

TimerLeft.Enabled = False

TimerRight.Enabled = False

End If

End Sub

Private Sub CmdDown_Click()

TimerUp.Enabled = False

TimerDown.Enabled = True

TimerLeft.Enabled = False

TimerRight.Enabled = False

End Sub

Private Sub CmdLeft_Click()

TimerUp.Enabled = False

TimerDown.Enabled = False

TimerLeft.Enabled = True

TimerRight.Enabled = False

End Sub

Private Sub CmdNew_Click()

ShpCircle.Left = Int(Rnd * 154 + 16)

ShpSquare.Left = Int(Rnd * 154 + 16)

ShpCircle.Top = Int(Rnd * 66 + 48)

ShpSquare.Top = Int(Rnd * 66 + 48)

Label1.Caption = ""

df = 0

dmin = Abs(ShpCircle.Left - ShpSquare.Left) + Abs(ShpCircle.Top - ShpSquare.Top)

 End Sub

Private Sub CmdPight_Click()

TimerUp.Enabled = False

TimerDown.Enabled = False

TimerLeft.Enabled = False

TimerRight.Enabled = True

End Sub

Private Sub CmdUp_Click()

TimerUp.Enabled = True

TimerDown.Enabled = False

TimerLeft.Enabled = False

TimerRight.Enabled = False

End Sub

Private Sub Form_Load()

Randomize

ShpCircle.Left = Int(Rnd * 154 + 16)

ShpSquare.Left = Int(Rnd * 154 + 16)

ShpCircle.Top = Int(Rnd * 66 + 48)

ShpSquare.Top = Int(Rnd * 66 + 48)

End Sub

Private Sub TimerDown_Timer()

ShpCircle.Top = ShpCircle.Top + 1

df = df + 1

Call Check

End Sub

Private Sub TimerLeft_Timer()

ShpCircle.Left = ShpCircle.Left - 1

df = df - 1

Call Check

End Sub

Private Sub TimerRight_Timer()

ShpCircle.Left = ShpCircle.Left + 1

df = df + 1

Call Check

End Sub

Private Sub TimerUp_Timer()

ShpCircle.Top = ShpCircle.Top - 1

df = df - 1

Call Check

End Sub

Проект МОЗАИКА

Постановка задачи.

Проект предназначен для создания узоров, симметричных относительно четырех осей симметрии. В том месте, где находится указатель курсора, рисуется круг с центром в точке с координатами (x,y). Для этого используется метод Circle. После этого программа создает ещё семь кругов так, чтобы на форме получился узор, симметричный относительно четырех осей. На рисунке 1 показаны оси симметрии, точка А с координатами (x,y), координаты семи точек, образующих симметричный узор, а также пример узора Рисунок 2.
Координаты точек, указанные на рисунке, справедливы только в том случае, если система координат, связанная с формой, имеет начало в середине формы.


Метод Scale устанавливает систему координат, связанную с формой или объектом PictureBox в соответствии с желанием программиста.


Формат:


<Объект>.Scale (x1,y1) – (x2,y2)


где  

Объект – имя формы или объекта PictureBox для которого определяется система координат;

(x1,y1) – числа или арифметические выражения, задающие устанавливаемые координаты верхнего левого угла объекта. По умолчанию (если в проекте не использован метод Scale), координаты верхнего левого угла равны (0,0);
(x2,y2) – числа или арифметические выражения, задающие устанавливаемые координаты нижнего правого угла объекта. По умолчанию (если в проекте не использован метод Scale), первая координата нижнего правого угла равна ширине объекта, а вторая – высоте в тех единицах измерения, которые установлены свойством ScaleMode.

Система координат, которая не зависит от размера формы, начало координат всегда располагается в середине формы задается следующим образом:

Form1.Scale (-Form1.ScaleWidth / 2, Form1.ScaleHeight / 2)-(Form1.ScaleWidth / 2, -Form1.ScaleHeight / 2)

В проекте используется событие MouseMove – в момент перемещения мышки над объектом
Private Sub Form_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)
Button – переменная, которая показывает, какая кнопка мышки нажата:

1, если нажата левая кнопка;

2, если нажата правая кнопка;

3, если нажата средняя кнопка мыши;
 Shift – переменная, которая показывает, были ли на клавиатуре нажаты Shift, Ctrl или Alt в момент наступления события;

1, если нажата Shift;
2, если нажата Ctrl;
3, если нажата Alt;
 X и Y – переменные, определяющие координаты курсора в момент наступления события.

Код процедуры рисования мозаики.

Private Sub Form_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)

red = Rnd * 155 + 100

green = Rnd * 155 + 100

blue = Rnd * 55 + 100

Form1.FillColor = RGB(red, green, blue)

If Button = 1 Then

Circle (X, Y), 5, RGB(0, 100, 0)

Circle (-X, Y), 5, RGB(0, 100, 0)

Circle (X, -Y), 5, RGB(0, 100, 0)

Circle (-X, -Y), 5, RGB(0, 100, 0)

Circle (Y, X), 5, RGB(150, 100, 0)

Circle (Y, -X), 5, RGB(150, 100, 0)

Circle (-Y, X), 5, RGB(150, 100, 0)

Circle (-Y, -X), 5, RGB(150, 100, 0)

End If

End Sub 
Пользовательское меню. Menu Editor.

 Большинство приложений Windows имеют в верхней части окна панель, на которой расположен набор команд. Эту панель называют Меню команд.

В проекте создано меню, которое содержит четыре команды:

· Очистка экрана (стирает рисунок с формы);

· Печать (выводит рисунок на принтер);

· Система координат (устанавливает начало системы координат в середину формы после изменения размера формы во время работы программы);

· Выход (выход из программы).

Для вызова редактора меню необходимо выполнить одно из ниже предложеннго 

· Инструменты( Редактор меню

·  Ctrl+E

·  Кнопка на стандартной панели

 
Команды меню реагируют на событие Click так же, как обычные командные кнопки.

Private Sub MnuCls_Click()’ очистка формы
Cls

Form1.BackColor = RGB(Rnd * 255, Rnd * 255, Rnd * 255)

End Sub

Private Sub MnuCoord_Click()’задание системы координат
Form1.Scale (-Form1.ScaleWidth / 2, Form1.ScaleHeight / 2)-(Form1.ScaleWidth / 2, -Form1.ScaleHeight / 2)

End Sub

Private Sub MnuEnd_Click()’ выход из программы
End

End Sub

Private Sub MnuPrint_Click()’ вывод рисунка на печать
Form1.PrintForm

End Sub 

Бегущая строка

Dim n As Integer

Private Sub Timer1_Timer()

'Бегущая строка
n = n + 2

Label1.Left = Label1.Left + n

If Label1.Left > Form1.Width + Form1.Left Then

n = 0

Label1.Left = 0 - Label1.Width

Label1.ForeColor = RGB(Rnd * 255, Rnd * 255, Rnd * 255)

End If

End Sub

В данном проекте в качестве «бегущей строки» выступает объект Label1 - метка, у которого изменяется свойство Left.

Исследуйте модель. Какими способами можно изменить скорость бегущей строки? Как изменить направление движения строки?  
Ниже предлагается второй способ создания бегущей строки. 
Текст строки создается в графическом редакторе Paint и сохраняется в папке с проектом. На форме размещаются два объекта Picture1 и Picture2, объект Timer1 и кнопка Command1 для завершения работы программы.

Исследуйте компьютерную модель. Попробуйте самостоятельно разобраться в коде проекта и ответить на вопросы: Зачем необходим объект Picture2? Какую роль он играет в проекте? 
Какие свойства задаются объекту Picture2 при инициализации формы?

   Picture2.Visible = False

   Picture2.AutoSize = True

   Picture2.Picture = LoadPicture(CurDir + "\1.bmp")




Dim fW As Integer, fH As Integer    ' ширина и высота картинки
Dim x As Integer, y As Integer
' щелчок на кнопке Ok

Private Sub Command1_Click()

   Unload Form1

End Sub
' инициализация формы
Private Sub Form_Initialize()

   Picture2.Visible = False

   Picture2.AutoSize = True

   Picture2.Picture = LoadPicture(CurDir + "\1.bmp")

   fW = Picture2.Width

   fH = Picture2.Height

   Form1.Height = Form1.Height + Picture2.Height - _

                  Picture1.Height

   Command1.Top = Command1.Top + Picture2.Height - _

                  Picture1.Height

   Picture1.Height = Picture2.Height

   x = 0

   y = 0

   Timer1.Interval = 50

   Timer1.Enabled = True

End Sub

' обработка сигнала таймера

Private Sub Timer1_Timer()

   x = x + 40

   ' вывод части Picture1 в Picture2

   Picture1.PaintPicture Picture2.Picture,    -x, -y

   ' если выведенная часть картинки не полностью

   ' покрывает область вывода (Picture2), дорисовываем

   ' часть картинки рядом справа
   If x + Picture1.ScaleWidth >= Picture2.ScaleWidth Then

      Picture1.PaintPicture Picture2.Picture, _

      -x + Picture2.ScaleWidth, -y

   End If

   If x >= Picture2.Width Then x = 0

End Sub

Титры
Dim fW As Integer, fH As Integer    ' ширина и высота картинки
Dim x As Integer, y As Integer


' щелчок на кнопке Ok
Private Sub Command1_Click()

   Unload Form1

End Sub
' инициализация формы
Private Sub Form_Initialize()

   Picture2.Visible = False

   Picture2.AutoSize = True

   Picture2.Picture = LoadPicture(CurDir + "\banner.bmp")

   fW = Picture2.Width

   fH = Picture2.Height

   Form1.Width = Form1.Width + Picture2.Width - Picture1.Width

   Picture1.Width = Picture2.Width

   x = 0

   y = 0

   Timer1.Interval = 50

   Timer1.Enabled = True

End Sub

' обработка сигнала таймера

Private Sub Timer1_Timer()

   y = y + 10

   ' вывод части Picture1 в Picture2

   Picture1.PaintPicture Picture2.Picture, _

   -x, -y

   ' если выведенная часть картинки не полностью

   ' покрывает область вывода (Picture2), дорисовываем

   ' верхнюю часть картинки снизу
   If y + Picture1.ScaleHeight >= Picture2.ScaleHeight Then

      Picture1.PaintPicture Picture2.Picture, _

      -0, -y + Picture2.ScaleHeight

   End If

   If y >= Picture2.ScaleHeight Then y = 0

End Sub

Проект Часы
Исследуйте компьютерную модель часы.

Const R = 60            ' радиус циферблата часов
Const GRAD = 0.0174532  ' коэффициент пересчета угла

                        ' из градусов в радианы

Const PI = 3.14159265

Dim x0 As Integer, y0 As Integer    ' центр циферблата
Dim ahr As Integer                  ' положение стрелок (угол)

Dim amin As Integer

Dim asec As Integer

' инициализация формы
Private Sub Form_Initialize()

   Form1.Height = (Form1.Height - Form1.ScaleHeight) + _

                  (R + 30) * 2 * Screen.TwipsPerPixelY

   Form1.Width = (Form1.Width - Form1.ScaleWidth) + _

                 (R + 30) * 2 * Screen.TwipsPerPixelX

   x0 = R + 30

   y0 = R + 30

   ' положение стрелок
   ahr = 90 - Hour(Time) * 30 - (Minute(Time) / 12) * 6

   amin = 90 - Minute(Time) * 6

   asec = 90 - Second(Time) * 6

   Timer1.Interval = 1000   ' период сигналов таймера - 1 сек

   Timer1.Enabled = True    ' запуск таймера
   Form1.ScaleMode = 3

End Sub

' процедура вычерчивает вектор заданной длины из точки (x0,y0)

Sub Vector(x0 As Integer, y0 As Integer, _

           a As Integer, l As Integer)

   ' x0, y0 - начало вектора

   ' a - угол между осью X и вектором

   ' l - длина вектора

   Dim x, y As Integer  ' координаты конца вектора

   x = Round(x0 + l * Cos(a * GRAD))

   y = Round(y0 - l * Sin(a * GRAD))

   Line (x0, y0)-(x, y)

End Sub

' процедура рисует стрелки

Sub DrawClock()

   ' шаг секундной и минутной стрелок 6 градусов,

   ' часовой - 30.

   ' стирание изображений стрелок

   Form1.DrawWidth = 3   ' задание толщины линии

   Form1.ForeColor = Form1.BackColor

   ' часовая стрелка
   Call Vector(x0, y0, ahr, R - 20)

   ' минутная стрелка
   Call Vector(x0, y0, amin, R - 15)

   ' секундная стрелка
   Call Vector(x0, y0, asec, R - 7)

   ' определение нового положения стрелок

   ahr = 90 - Hour(Time) * 30 - (Minute(Time) / 12) * 6

   amin = 90 - Minute(Time) * 6

   asec = 90 - Second(Time) * 6

   ' прорисовка стрелок на новом положении

   ' часовая стрелка

   Form1.DrawWidth = 3

   Form1.ForeColor = RGB(0, 0, 0)

   Call Vector(x0, y0, ahr, R - 20)

   ' минутная стрелка

   Form1.DrawWidth = 2

   Call Vector(x0, y0, amin, R - 15)

   ' секундная стрелка

   Form1.DrawWidth = 1

   Form1.ForeColor = RGB(200, 0, 0)

   Call Vector(x0, y0, asec, R - 7)

End Sub

' прорисовка циферблата и начальных стрелок

Private Sub Form_Paint()

   Dim x As Integer     ' координаты маркера
   Dim y As Integer     ' на циферблате
   Dim a As Integer     ' угол между осью X и

                        ' прямой (x0, y0) - (x, y)

   Dim h As Integer     ' метка часовой риски

   Form1.DrawWidth = 1

   Form1.ForeColor = RGB(0, 0, 0)

   a = 0    ' метки ставим от 3-х часов, против

            ' часовой стрелки

   h = 3    ' угол 0 градусов - это 3 часа

   ' циферблат
   While (a < 360)

      x = x0 + Round(R * Cos(a * 2 * PI / 360))

      y = x0 - Round(R * Sin(a * 2 * PI / 360))

      If (a Mod 30) = 0 Then

         Circle (x, y), 2

         ' вывод цифр по большему радиусу

         CurrentX = x0 + Round((R + 15) * _

                    Cos(a * 2 * PI / 360)) - 7

         CurrentY = x0 - Round((R + 15) * _

                    Sin(a * 2 * PI / 360)) - 7

         Print h

         h = h - 1

         If h = 0 Then h = 12

      Else: Circle (x, y), 1

      End If

      a = a + 6         ' 1 минута - 6 градусов
   Wend

   Call DrawClock

End Sub

' прорисовка текущих положений стрелок часов

Private Sub Timer1_Timer()

   Call DrawClock

End Sub

6. Конец





a  = 0





да








4. a = 7?





3. Увеличить на единицу номер элемента массива Image2


a = a + 1











2. Поместить в объект Image1 рисунок из элемента массива Image2 с номером a.


Image1.Picture = Image1(a).Picture








1. Начало























Массив объектов Image2(k)





Объект Image1

Кнопка для реализации выхода из проекта





Объект Picture1 – вся видимая часть





Объект Picture2 –дорога, нарисованная в графическом редакторе PHOTOSHOP





Объект Picture3 – машина, подготовленная в графическом редакторе PHOTOSHOP

Объект  Picture3





Timer1 для реализации движения синего экскаватора





Timer2 для реализации движения красного экскаватора.





Timer3 для выбора случайных шагов перемещения влево красного и синего экскаваторов.

